Molecular Coding Format manual

by Akira Yamaji 2015.9.25

No.1 Chain (1)

```
plus(+):anticlockwize,minus(-)clockwize
bold arrow is default angle and position

  <30,-60,60,-60,60,-60,60
```



No.2 Chain (2)

```
!,: take 60 or -60 depend on
      current angle and enviroment

  !6

  ** !6: !,!,!,!,!,!
```



No.3 Ring

```
six membered ring

  ?6

  ** ?6 : <-120,60,60,60,60,60,&1
     &1 : make bond connect to A1
```



No.4 Rotate current angle

```
<angle:rotate current angle

   <30,!4,<30,?6
```



No.5 Change bond type (1)

```
~bond change bond
  dm:double,dl:double(left),dr:double(right),
  wf:wedge forward,wb:wedge backward,
  zf:wedge dotted,zb:wedge dotted backward

  <30,!~dm,!~dl,!~dr,!~wf,!~wb,!~zf,!~zb
```



No.6 Change bond type (2)

```
  Bn=bond type : change bond type at Bn
  vf:vector forward, vb:vector backward,
  tm:triple, wv:waved, bd:broad single,
  ov:over line

  <30,!9,1=vf,2=vb,3=tm,4=dt,5=wv,
  @(7~bd^-45‘2,9~ov^45‘2)/Me
```



No.7 **Change bond length (1)**

```
Bn‘length : change bond length at Bn

  <30,!2,!2‘1.2,!2
```



No.8 **Change bond length (2)**

```
‘‘length : change all bond length after

  <30,!2,‘‘1.2,!4
```



No.9 **Change chain length (3)**

```
?n‘length : change ring length

  <30,?6,3\,|,?6‘1.2,4=?6
```



No.10 **Change atom (1)**

```
Insert hetero atom

  <30,!2,O,!2
```



No.11 **Change atom (2)**

```
2:O : change A2 C to O
@(3,5)N : change  A3,A5 C to N

  <30,!6,2:O,@(3,5)N

An(n:numeric): atom number
```



No.12 **Change atom (3)**

```
2:N : change A2 C to N

  ?6,4\,?6,2:N
```



No.13 **Change atom (brock address)**

```
2:N : change A2(brock address) C to N

?6,4\,|,?6,2:N

| : divide brock
```



No.14 **Change atoma (absolute address)**

```
#2:N : change A#2 C to N

  ?6,4\,|,?6,#2:N
```



No.15 **Change atom (relative address)**

```
-2:N : change A(-2) C to N

  ?6,4\,?6,-2:N
```



No.16 **Fuse ring (attached 1 bond)**

```
?6,3=?5 : fuse ?5 at B3

  ?6,3=?5

Bn(n:numeric): bond number
```



No.17 **Fuse multi ring (attached 1 bond)**

```
?6,$(-3,-4,-4,-2,-2,-4,-4)?6
?6,$(4,8,13,20,25,28,33)?6
```



No.18 **Fuse ring (attached 2 bond)**

```
(4,11)=?6[4]  : fuse 4/6 ring to B11..B4
(4,11)=?5[3]  : fuse 3/5 ring to B11..B4
(4,11)=?4[2]  : fuse 2/4 ring to B11..B4

MCd(1,.7)( 0,0)(<30,?6,3=?6,(11,4)=?6[4])
MCd(1,.6)(.5,1)(<30,?6,3=?6,(11,4)=?5[3])
MCd(1,.6)( 1,0)(<30,?6,3=?6,(11,4)=?4[2])
```



No.19 **Fuse ring (attached 3,4 bond)**

```
(16,4)=?6[3] : fuse 3/6 ring to B16..B4
(16,4)=?5[2] : fuse 2/5 ring to B16..B4
(21,3)=?6[2] : fuse 2/6 ring to B21..B3

MCd(1,.55)( 0, 0)(?6,$(3,10)?6,(16,4)=?6[3])
MCd(1,.55)(.43,1)(?6,$(3,10)?6,(16,4)=?5[2])
MCd(1,.53)(1,0)(<30,?6,$(2,10,15)?6,(21,3)=?6[2])
```



No.20 **Spiro ring**

```
4*,?5 : add ?5(5 membered ring) at A4

<30,!6,4*,?5

An* : jump to An
```



No.21 **Branch bond (1)**

```
2\  : 2*,0     4*\ : 4*,0~wf
6\* : 6*,0~zf  8\\ : 8*,0~dm

MCf(<30,!8,2\,!,4*\,!,6\*,!,8\\,!)
```



No.22 **Branch bond (2)**

```
2\~dr  : 2*,0~dr
4\‘1.5 : 4*,0‘1.5
6\^15 : 6*,0^15

MCf(<-30,!6,2\~dr,!,4\‘1.5,!,6\^15,-60)
```



No.23 **Insert substituent(1)**

```
MCf(<30,
!,/Me,!,/Et,!3,/Pr,!,/iPr,!3,/tBu,!,/Ph^-30,!)

** Me:methyl Et:ethyl Pr:propyl iPr:isopropyl
   tBu:tertial buthyl Ph:phenyl
```



No.24 **Insert substituent(2)**

```
/  : single          // : double
*/ : wedge forward   /* : wedge backward
** : direct

MCf(<30,!,//O,!,/*H,!,*/H,!,/?3,!,**?3,!)
```



No.25 Insert substituent(3)

```
~,^,‘,> : change type,angle,length,enviroment
          of substituent

MCf(<30,!‘1,/Me~zf‘2^30,!‘1,/Me~zf‘2^30,
    !2‘1,*/Pr>lr,!2‘1,*/Pr>rl,!‘1)
```



No.26 Add substituent(1)

```
sw_numberA:=1; numberA_end:=_skeletonA_end;
MCf(<30,!11,
     2:/Me,3:/Et,6:/Pr,7:/iPr,10:/tBu,11:/Ph^-30)

** _skeletonA_end :
   end number of molecular skeleton atom
```



No.27 Add substituent(2)

```
~,^,‘,> : change type,angle,length,enviroment
          of substituent

MCf(<30,?6,
    @(3‘2^30,3~wf,4~zf,6~wf^-30,6~zf^30)/H)
```



No.28 Add substituent(3)

```
~,^,‘,> : change type,angle,length,enviroment
          of substituent

MCF(<30,!7‘1,
@(2,3)/*Me‘2^30,5:*/Pr>lr,7:*/Pr>rl)
```



No.29 Chain strech direction environment (1)

```
>hz : horizontal enviroment (default)
>vt : vertical enviroment

  ?4,
  @(3^-90,3^-30,3^90)/’(!3,"{>hz}")>hz,
  @(1^-60,1‘1.5,1^60)/’(!2,"{>vt}")>vt
```



No.30 Chain strech direction environment (2)

```
>lr : left-right enviroment
>rl : right-left enviroment

  <30,!6,
  @(3^-30,3,3^30)/’(!3,"{>lr}")>lr,
  @(5^-30,5,5^30)/’(!3,"{>rl}")>rl
```



No.31 Chain strech direction environment (3)

```
>45 : fixed angle enviroment
>'(-90,90,-90) : multi angle enviroment

 <-30,!6,@(2>45)/'(!3,"{>45}"),
 @(6>'(-90,90,-90))/'(!2,"{>(-90,90,-90)}")
```



No.32 Change atom and Substituent

```
NH,SOO : inset hetero atom and substituent
         simultaneously

 <30,!3,NH,!,SOO,!3
```



No.33 Change color

```
@(5)green : change color of A5 green
$(3)red :   change color of B3 red

 <30,Ph,@(2,5)N,
        2:red,5:blue,3=green
```



No.34 Chain start multiple characters

```
if chain start multi charactor string,
use !0 instead of !

 MCd( 1,1)(0  ,0.9)(<30,COOH,!0,!3,COOH)

 MCd(.8,1)(0.3,0.1)(<30,COOH,!4,COOH)
```



No.35 User definition

```
iBuOH : user defined substructure

 iBuOH:='(!,/Me,!,OH)
 MCf(<30,?6,@(4,6)/iBuOH)
```



No.36 Inline definition

```
Insert user defined substructure

 <30,!3,/'(!,/Me,!,OH),!3
```



No.37 **Example(1) Ampicillin**

```
<45,?4,2:N,2=?5,-1:S,
@(3^45,4^-45)/*H,1://O^15,
5:/*COOH^-18,@(6^35,6^-35)/Me,
4\^75,NH,!,//O,!,NH,!,Ph
```



No.38 **Example(2) Cholesterol**

```
<30,?6,$(-4,-2)?6,-4=?5,7=dl,
1:*/OH,@(4,12)*/Me^60,9:*/H^60,
10:/*H^180,@(11,-1)/*H^-60,
-1\^17,/*Me,!4,/Me,!,
```



No.39 **Example(3) alfa-Tocopherol**

```
<30,Ph,3=?6,
7:O,@(1,2,5)/Me,8:/*Me^60,6:/OH,
8\,|,!12,@(4,8)/*Me,12:/Me
```



No.40 **Example(4) Caffeine**

```
<30,?6,-4=?5,$(3,8)dl,
@(2,6,7,9)N,@(2,6,9)/Me,
@(1,5)//O
```



No.41 **Example(5) Lycorine**

```
<30,Ph,-4=?6,-2=?6,6=?5,(9,12)=?5[3],
13=dl,8:N,@(15,17)O,
9:/*H^180,10:*/H^60,13:*/OH,14:/*OH
```



No.42 **Example(6) Gibberellin**

```
<12,?6'1.3,3=?5,9=?7,12\^160'1.6,&8,
4\^155~zf'1.2,O,55,//O^180'1,&2~zb,
5=dl,11=wf,13=wb,7:/COOH,11://Me,1:*/OH,
12:/*OH,2:*/Me,@(3^-60,9^60)*/H
```



No.43 **Font size**

```
beginfont("EN:Caffeine")
  font_wd#:=30mm#;  %<==font width
  font_ht#:=20mm#;  %<==font height
  sw_font_frame:=1;
  MCf(<30,?6,-4=?5,$(3,8)dl,@(2,6,7,9)N,
     @(2,6,9)/Me,@(1,5)//O)
endfont
```



No.44 **Max ratio bond/width length**

```
max_bond_width:=0.10;
MCd(1,1)( 0, .5)(<30,Ph)
max_bond_width:=0.15;          %<<== default
MCd(1,1)(.33,.5)(<30,Ph)
max_bond_width:=0.30;
MCd(1,1)( 1, .5)(<30,Ph)
```



No.45 **Ratio thickness/bond length**

```
ratio_thickness_bond:= 0.005;
MCd(1,.6)(0, .5)(<30,Ph)
ratio_thickness_bond:= 0.015;    %<<== default
MCd(1,.6)(.5,.5)(<30,Ph)
ratio_thickness_bond:= 0.030;
MCd(1,.6)(1, .5)(<30,Ph)
```



No.46 **Offset thickness of bond**

```
beginfont() offset_thickness#:=0pt#;
MCd(1,.3)(0, .5)(<30,Ph) endfont
beginfont() offset_thickness#:=0.2pt#; %<<== default
MCd(1,.3)(.5,.5)(<30,Ph) endfont
beginfont() offset_thickness#:=0.5pt#;
MCd(1,.3)(1, .5)(<30,Ph) endfont
```



No.47 **Ratio char/bond thickness**

```
ratio_char_bond:=1.0;
MCd(1,.6)(0, .5)(<30,?6,5:O,2:NH)
ratio_char_bond:=1.5;          %<<== default
MCd(1,.6)(.5,.5)(<30,?6,5:O,2:NH)
ratio_char_bond:=2.0;
MCd(1,.6)(1, .5)(<30,?6,5:O,2:NH)
```



No.48 **Ratio bondgap/bond length**

```
ratio_bondgap_bond:= 0.10;
MCd(1,.6)(0, .5)(<30,Ph)
ratio_bondgap_bond:= 0.15;     %<<== default
MCd(1,.6)(.5,.5)(<30,Ph)
ratio_bondgap_bond:= 0.20;
MCd(1,.6)(1, .5)(<30,Ph)
```



No.49 **Offset of doublebond gap**

```
beginfont() offset_bond_gap#:=0.0pt#;
MCd(1,.3)(0, .5)(<30,Ph) endfont
beginfont() offset_bond_gap#:=0.3pt#; %<<== default
MCd(1,.3)(.5,.5)(<30,Ph) endfont
beginfont() offset_bond_gap#:=1.0pt#;
MCd(1,.3)(1, .5)(<30,Ph) endfont
```



No.50 **Ratio atom/bond length**

```
ratio_atom_bond:= 0.25;
MCd(1,.6)(0, .5)(<30,?6,@(2,5)O)
ratio_atom_bond:= 0.36;          %<<== default
MCd(1,.6)(.5,.5)(<30,?6,@(2,5)O)
ratio_atom_bond:= 0.45;
MCd(1,.6)(1, .5)(<30,?6,@(2,5)O)
```



No.51 **Offset of atom width**

```
beginfont() offset_atom#:=0.0pt#;
MCd(1,.3)(0, .5)(<30,Ph,@(2,4,6)N) endfont
beginfont() offset_atom#:=0.8pt#; %<<== default
MCd(1,.3)(.5,.5)(<30,Ph,@(2,4,6)N) endfont
beginfont() offset_atom#:=2.0pt#;
MCd(1,.3)(1, .5)(<30,Ph,@(2,4,6)N) endfont
```



No.52 **Ratio wedge/bond length**

```
ratio_wedge_bond:=0.10;
MCd(1,.6)(0, .5)(<30,?6,5:*/Me)
ratio_wedge_bond:=0.12;          %<<== default
MCd(1,.6)(.5,.5)(<30,?6,5:*/Me)
ratio_wedge_bond:=0.20;;
MCd(1,.6)(1, .5)(<30,?6,5:*/Me)
```



No.53 **Offset of wedge width**

```
beginfont("EN:Offset_wedge")
offset_wedge#:=0.0pt#;
MCd(1,.3)(0, .5)(<30,?6,5:*/Me) endfont
beginfont() offset_wedge#:=0.4pt#;  %<<== default
MCd(1,.3)(.5,.5)(<30,?6,5:*/Me) endfont
beginfont() offset_wedge#:=1.0pt#;
MCd(1,.3)(1, .5)(<30,?6,5:*/Me) endfont
```



No.54 **Ratio font atom gap/atom length**

```
ratio_atomgap_atom:=0.0;
MCd(1,.6)(0, .5)(<30,?6,@(2,5)O)
ratio_atomgap_atom:=0.050;       %<<== default
MCd(1,.6)(.5,.5)(<30,?6,@(2,5)O)
ratio_atomgap_atom:=0.12;
MCd(1,.6)(1, .5)(<30,?6,@(2,5)O)
```



No.55 Ratio chain/ring length

```
ratio_chain_ring:= 0.4;
MCd(1,.6)(0, .5)(<30,?6,5:/Et)
ratio_chain_ring:= 0.66;        %<<== default
MCd(1,.6)(.5,.5)(<30,?6,5:/Et)
ratio_chain_ring:= 1.0;
MCd(1,.6)(1, .5)(<30,?6,5:/Et)
```



No.56 Ratio zebra gap/bond length

```
ratio_zebragap_bond:=0.06;
MCd(1,.6)(0, .5)(<30,Ph,5:/*Me‘1)
ratio_zebragap_bond:=0.12;      %<<== default
MCd(1,.6)(.5,.5)(<30,Ph,5:/*Me‘1)
ratio_zebragap_bond:=0.20;
MCd(1,.6)(1, .5)(<30,Ph,5:/*Me‘1)
```



No.57 Margin left and right

```
margin_left_right:=0mm;
MCd(1,1)(0.5,0.9)(<30,CH3,!0,!17,CH3)
margin_left_right:=0.4mm;       %<<== default
MCd(1,1)(0.5,0.5)(<30,CH3,!0,!17,CH3)
margin_left_right:=5mm;
MCd(1,1)(0.5,0.1)(<30,CH3,!0,!17,CH3)
```



No.58 Margin top and bottom

```
margin_top_bottom:=0mm;
MCd(1,1)(0.1,0.5)(<30,Ph,2:/OH,5:/NH2)
margin_top_bottom:=0.4mm;       %<<== default
MCd(1,1)(0.5,0.5)(<30,Ph,2:/OH,5:/NH2)
margin_top_bottom:=5mm;
MCd(1,1)(0.9,0.5)(<30,Ph,2:/OH,5:/NH2)
```



No.59 Switch Start Vector

```
MCd(1,.8)( 0,0.5)(<30,Ph,4:/Cl,3:/F)
sw_start_vector:=1;
MCd(1,.8)( 1,0.5)(<30,Ph,4:/Cl,3:/F)
```



No.60 Switch Numbering atom

```
numberA_start:=3; numberA_end:=8;

sw_numberA:=1; MCd(1,1)(.5,.9)(<30,!9)
sw_numberA:=2; MCd(1,1)(.5,.5)(<30,!9)
sw_numberA:=3; MCd(1,1)(.5,.1)(<30,!9)

** default: numberA_start=1 numberA_end=4095
```



No.61 **Switch Numbering bond**

```
numberB_start:=3; numberB_end:=8;

sw_numberB:=1; MCd(1,1)(.5,.9)(<30,!9)
sw_numberB:=2; MCd(1,1)(.5,.5)(<30,!9)
sw_numberB:=3; MCd(1,1)(.5,.1)(<30,!9)

** default: numberB_start=1 numberB_end=4095
```



No.62 **Switch font frame**

```
sw_font_frame:=1;

MCf(<30,Ph,4:/Cl,3:/F)
```



No.63 **Switch molecular frame**

```
MCd(1,.8)(0,0.5)(<30,Ph,4:/Cl,3:/F)
sw_mol_frame:=1;
MCd(1,.8)(1,0.5)(<30,Ph,4:/Cl,3:/F)
```



No.64 **Switch atom frame**

```
sw_atom_frame:=1;

MCf(<30,Ph,4:/COOH,3:/OH)
```



No.65 **Switch solid mode**

```
MCd(1,.8)( 0,0.5)(<30,Ph,4:/Cl,3:/F)
sw_solid:=1; ratio_bond_width:=0.08;
MCd(1,.8)( 1,0.5)(<30,Ph,4:/Cl,3:/F)
```



No.66 **Switch Expand**

```
MCd(1,.6)(0,0.5)(<30,Ph,4:/COOH,3:/NH2)
sw_expand:=1;
MCd(1,.6)(1,0.5)(<30,Ph,4:/COOH,3:/NH2)
```



No.67 Function MCd (draw)

```
MCd(a,b)(c,d)(...)
  a: ratio molecular width/font width
  b: ratio molecular hight/font hight
  c: x axis position  d:y axis position

MCd(1,0.8)(0.2,0.9)(<30,Ph,2:/OH,5:/NH2)
MCd(1,0.8)(0.8,0.1)(<30,Ph,2:/OH,5:/NH2)
```



No.68 Function MCf (fit draw)

```
MCf(...) : MCd(1,1)(0.5,0.5)(...)

sw_font_frame:=1;  sw_mol_frame:=1;

MCf(<30,Ph,2:/OH,5:/NH2)
```



No.69 Local setting

```
beginfont("NO:1")
  sw_mol_frame:=1; % <<== Local setting
  MCd(1,.5)(0.0,0.5)(<30,Ph,@(2)N) endfont
beginfont("NO:2")
  MCd(1,.5)(0.5,0.5)(<30,Ph,@(2,4)N) endfont
beginfont("NO:3")
  MCd(1,.5)(1.0,0.5)(<30,Ph,@(2,4,6)N) endfont
```



No.70 Global setting

```
sw_mol_frame:=1; % <<== Global setting
beginfont("NO:1")
  MCd(1,.5)(0.0,0.5)(<30,Ph,@(2)N) endfont
beginfont("NO:2")
  MCd(1,.5)(0.5,0.5)(<30,Ph,@(2,4)N) endfont
beginfont("NO:3")
  MCd(1,.5)(1.0,0.5)(<30,Ph,@(2,4,6)N) endfont
```



No.71 Output molecular infomation

```
beginfont() sw_info_weight:=sw_info_formula:=1;
MCf(...) endfont
%% Output to mcf_man_soc-info.aux %%
\INFO{{F:mcf_man_soc}{C:82}{MWc:194.19174}
      {FMc:C8H10N4O2}}%
MWc:calculated molecular weight
FMc:calculated molecular formula
```



No.72 Output additional infomation

```
beginfont("EN:Caffeine","CAS:58-08-2")
MCf(...) endfont
%% Output to mcf_man_soc-info.aux %%
\INFO{{F:mcf_man_soc}{C:83}{EN:Caffeine}
     {CAS:58-08-2}}%
*F:filename *C:char number EN:molecular name
CAS:CAS number *:default output
```